


**California Wildlife Habitat Relationships Program
California Department of Fish and Game**

**HABITAT SUITABILITY MODELS FOR USE WITH ARC/INFO:
MOUNTAIN CHICKADEE**



CWHR Technical Report No. 9
Sacramento, CA
June 1995

June 1995

HABITAT SUITABILITY MODELS FOR USE WITH ARC/INFO:
MOUNTAIN CHICKADEE

by

Irene C. Timossi

Ellen L. Woodard

Reginald H. Barrett

Department of Environmental Science, Policy, and Management

University of California

Berkeley, CA 94720

and the

Sierra Nevada Ecosystem Project

California Wildlife Habitat Relationships Program

Wildlife Management Division

California Department of Fish and Game

1416 Ninth Street

Sacramento, CA 95814

Suggested Citation: Timossi, I. C., E. L. Woodard, and R. H. Barrett. 1995. Habitat suitability models for use with ARC/INFO: Mountain chickadee. Calif. Dept. of Fish and Game, CWHR Program, Sacramento, CA. CWHR Tech. Report No. 9. 23 pp.

TABLE OF CONTENTS

PREFACE	ii
ACKNOWLEDGMENTS	v
HABITAT USE INFORMATION	1
General	1
Food	1
Water	2
Cover	2
Reproduction	2
Interspersion and Composition	2
HABITAT SUITABILITY INDEX (HSI) MODEL	3
Model Applicability	3
Geographic area	3
Season	3
Cover types	3
Minimum habitat area	3
Verification level	4
Model Description	4
Overview	4
Cover component	4
Distance to water	5
Species' distribution	5
Spatial analysis	5
Definitions	5
Application of the Model	7
Problems with the Approach	8
Cost	8
Dispersal distance	9
Day to day distance	9
SOURCES OF OTHER MODELS	9
REFERENCES	10
APPENDIX 1: Mountain Chickadee Macro	12

ACKNOWLEDGMENTS

The primary credit for this document must go to the field biologists and naturalists that have published the body of literature on the ecology and natural history of this species. They are listed in the References section. Ecological information of this sort is generally very expensive and time-consuming to obtain. Yet this basic ecological understanding is exactly what is needed most if the goal of accurately predicting changes in distribution and abundance of a particular species is ever to be achieved. The CWHR System is designed to facilitate the use of existing information by practicing wildlife biologists. We hope it will also stimulate funding for basic ecological research. Funding for producing this model was provided by the California Department of Forestry and Fire Protection and the University of California Agricultural Experiment Station.

We thank Barry Garrison, Karyn Semka, and Sandie Martinez of the California Department of Fish and Game for their assistance in typing, editing, and producing this report.

MOUNTAIN CHICKADEE (*Parus gambeli*)

HABITAT USE INFORMATION

General

The mountain chickadee (*Parus gambeli*) inhabits open, montane coniferous forests in northern British Columbia through the western United States and northern Baja California (Bevier 1988). In northern California, the mountain chickadee is a common to abundant resident in the Sierra Nevada, Cascade, Klamath, and interior Coast Ranges. In southern California, they are common residents of the coastal ranges from Santa Barbara to San Diego counties, and from the White to the New York mountains (Zeiner et al. 1990). The mountain chickadee frequents coniferous forests, especially ponderosa pine (*Pinus ponderosa*), Jeffrey pine (*P. jeffreyi*), and lodgepole pine (*P. contorta* var. *murrayana*), as well as red fir (*Abies magnifica*) and mixed conifer habitats (Zeiner et al. 1990). During the winter, individuals may move downslope into riparian and hardwood habitats. In the north, downslope movements may extend into the outer coastal ranges (Grinnell and Miller 1944; McCaskie et al. 1979; Garrett and Dunn 1981).

Food

Though mountain chickadees are primarily foliage-gleaning insectivores, but some seeds and other plant materials are consumed (Zeiner et al. 1990). In the Sierra Nevada, mountain chickadees foraged exclusively by gleaning, spending approximately 60% of the time foraging on foliage and the remainder on twigs and branches (Airola and Barrett 1985). Similarly, Brennan (1989) observed this species foraging almost exclusively by gleaning during the spring and summer, spending 60% of the time searching foliage, 34% searching twigs or small branches, 7% searching terminal buds, and 4% searching other substrates. During the winter, foraging techniques included both gleaning and flaking tree bark (Brennan 1989). Prey items consumed by nestling mountain chickadees in northern California included Hymenoptera (mainly larvae) (38.6%), larval Lepidoptera (25.0%), adult Coleoptera (12.6%), adult and larval Diptera (8.2%), spiders (5.6%), and adult Homoptera (3.2%) (Grundel and Dahlsten 1990). Incense cedar scale (*Xylocuculus macrocarpae*) may be used during the winter (Morrison et al. 1989).

Mountain chickadees forage in the foliage of conifers and, to a lesser extent, in deciduous trees. In southwestern Alberta, this species foraged at an average height of 10.6 m (34.6 ft) (Hill and Lein 1988), while in the central Sierra Nevada, chickadees foraged at heights ranging from 11-17 m (36-56 ft) (Brennan 1989). White fir (*Abies concolor*), incense cedar (*Calocedrus decurrens*), ponderosa pine, sugar pine (*P. lambertiana*), Douglas-fir (*Pseudotsuga menziesii*), and black oak (*Quercus kelloggii*) were commonly used for foraging by mountain chickadees in the central Sierra Nevada (Airola and Barrett 1985; Morrison et al. 1985), with pines and white fir being preferred (Airola and Barrett 1985). Mountain chickadees utilized incense cedar to a greater extent during the winter than during summer and black oak to a lesser degree during winter than during summer (Morrison et al. 1985).

Water

No water requirements have been documented for this species.

Cover

Mountain chickadees are found almost exclusively in coniferous forests, more rarely in mixed conifer-hardwood and riparian habitats (Grinnell and Miller 1944; Zeiner et al. 1990). In the Sierra Nevada, breeding mountain chickadees preferred closed canopy red fir stands, but they also used open canopy mixed conifer sites (Beedy 1981). Mountain chickadees, while utilizing both burned and unburned mixed coniferous forests stands on the eastside of the Sierra Nevada, were three times as abundant in the unburned area (Bock and Lynch 1970). In the central Sierra Nevada, mountain chickadees exhibited a preference for ponderosa pine, sugar pine, and white fir during the summer (Airola and Barrett 1985; Brennan 1989).

Reproduction

Depending upon local conditions, mountain chickadees begin breeding in early April to mid-May. Nests are concealed in old woodpecker holes, natural cavities in tree branches, trunks or stumps, or under rocks (Harrison 1978). They occasionally excavate or enlarge cavities of their own in rotten wood (Zeiner et al. 1990). Nests are frequently reused in subsequent years (Hill and Lein 1988). In southwestern Alberta, nests were located at an average height of 4.8 m (15.8 ft) above ground (Hill and Lein 1988). Occasionally nests are placed as high as 24 m (80 ft) (Harrison 1978). In the eastern Sierra, mountain chickadees preferred to nest in well decayed suitable fir snags averaging 8 m (26.2 ft) tall and 56 cm (22 in) dbh, with most nest being 3-4 m (9.8-13.1 ft) up the bole (Raphael and White 1984).

Interspersion and Composition

Mountain chickadees are found in small flocks from early autumn until the beginning of the next breeding season (Grinnell and Storer 1924). In winter the flocks often include other insectivorous species. In the Sierra Nevada, Morrison et al. (1987) found mountain chickadees flocking frequently with golden-crowned kinglets (*Regulus satrapa*) and red-breasted nuthatches (*Sitta canadensis*), and less often with brown creepers (*Certhia americana*), dark-eyed juncos (*Junco hyemalis*), ruby-crowned kinglets (*R. calendula*), and small woodpeckers and warblers.

In the Sierra Nevada, density estimates of nesting mountain chickadees range from a low of 67 birds/40 ha (100 ac) in open-canopy red fir forest to a high of 83 birds/40 ha (100 ac) in both open-canopy mixed conifer and closed-canopy red fir forests (Beedy 1981). Bock and Lynch (1970) found nesting densities of 5.2 pairs/40 ha (100 ac) and 15.6 pairs/40 ha (100 ac), respectively, in burned and unburned stands of mixed coniferous forests in the Sierra Nevada. In Lassen and Modoc Counties, densities of 38 pairs/40 ha (100 ac) were achieved by adding nest boxes (Dahlsten and Cooper 1979). Territories may be maintained by adults throughout the year (Dixon and Gilbert 1964; Dahlsten and Cooper 1979).

HABITAT SUITABILITY INDEX (HSI) MODEL

Model Applicability

Geographic area.

The California Wildlife Habitat Relationships (CWHR) System (Airola 1988; Mayer and Laudenslayer 1988; Zeiner et al. 1990) contains habitat ratings for each habitat type predicted to be occupied by mountain chickadees in California.

Season.

This model is designed to predict the suitability of habitat for mountain chickadees throughout the year. The model works best at predicting habitat suitability for breeding habitat.

Cover types.

This model can be used anywhere in California for which an ARC/INFO map of CWHR habitat types exists. The CWHR System contains suitability ratings for reproduction, cover, and feeding for all habitats mountain chickadees are predicted to occupy. These ratings can be used in conjunction with the ARC/INFO habitat map to model wildlife habitat suitability.

Minimum habitat area.

Minimum habitat area is defined as the minimum amount of contiguous habitat required before a species will occupy an area. Specific information on minimum areas required for mountain chickadees was not found in the literature. This model assumes two home ranges is the minimum area required to support a mountain chickadee population during the breeding season.

Verification level.

The spatial model presented here has not been verified in the field. The CWHR suitability values used are based on a combination of literature searches and expert opinion. We strongly encourage field testing of both the CWHR database and this spatial model.

Model Description

Overview.

This model uses CWHR habitat type as the main factor determining suitability of an area for this species.

A CWHR habitat type map must be constructed in ARC/INFO GRID format as a basis for the model. The GRID module of ARC/INFO was used because of its superior functionality for spatial modeling. Only crude spatial modeling is possible in the vector portion of the ARC/INFO program, and much of the modeling done here would have been impossible without the abilities of the GRID module. In addition to more sophisticated modeling, the GRID module's execution speed is very rapid, allowing a complex model to run in less than 30 minutes.

The following sections document the logic and assumptions used to interpret habitat suitability.

Cover component.

A CWHR habitat map must be constructed. The mapped data (coverage) must be in ARC/INFO GRID format. A grid is a GIS coverage composed of a matrix of information. When the grid coverage is created, the size of the grid cell should be determined based on the resolution of the habitat data and the home range size of the species with the smallest home range in the study. You must be able to map the home range of the smallest species with reasonable accuracy. However, if the cell size becomes too small, data processing time can increase considerably. We recommend a grid cell size of 30 m (98 ft). Each grid cell can be assigned attributes. The initial map must have an attribute identifying the CWHR habitat type of each grid cell. A CWHR suitability value is assigned to each grid cell in the coverage based on its habitat type. Each CWHR habitat is rated as high, medium, low or of no value for each of three life requisites: reproduction; feeding; and cover. The geometric mean value of the three suitability values was used to determine the base value of each grid cell for this analysis.

Distance to water.

No water requirement was found for this species.

Species' distribution.

The study area must be manually compared to the range maps in the CWHR Species Notes (Zeiner et al. 1990) to ensure that it is within the species' range. All grid cells outside the species' range have a suitability of zero.

Spatial analysis.

Ideally a spatial model of distribution should operate on coverages containing habitat element information of primary importance to a species. For example, in the case of woodpeckers, the size and density of snags as well as the vegetation type would be of great importance. For many small rodents the amount and size of dead and down woody material would be important. Unfortunately, the large cost involved in collecting microhabitat (habitat element) information and keeping it current makes it likely that geographic information system (GIS) coverages showing such information will be unavailable for extensive areas into the foreseeable future.

The model described here makes use of readily available information such as CWHR habitat type, elevation, slope, aspect, roads, rivers, streams and lakes. The goal of the model is to eliminate areas that are unlikely to be utilized by the species and lessen the value of marginally suitable areas. It does not attempt to address all the microhabitat issues discussed above, nor does it account for other environmental factors such as toxins, competitors or predators. If and when such information becomes available, this model could be modified to make use of it.

In conclusion, field surveys will likely discover that the species is not as widespread or abundant as predictions by this model suggest. The model predicts potentially available habitat. There are a variety of reasons why the habitat may not be utilized.

Definitions.

Home Range: the area regularly used for all life activities by an individual during the season(s) for which this model is applicable.

Dispersal Distance: the distance an individual will disperse to establish a new home range. In this model it is used to determine if Potential Colony Habitat will be utilized.

Day to Day Distance: the distance an individual is willing to travel on a daily or semi-daily basis to utilize a distant resource (Potential Day to Day Habitat). The distance used in the model is the home range radius. This is determined by calculating the radius of a circle with an area of one home range.

Core Habitat: a contiguous area of habitat of medium or high quality that has an area greater than two home ranges in size. This habitat is in continuous use by the species. The species is successful enough in this habitat to produce offspring that may disperse from this area to the Colony Habitat and Other Habitat.

Potential Colony Habitat: a contiguous area of habitat of medium or high quality that has an area between one and two home ranges in size. It is not necessarily used continuously by the species. The distance from a core area will affect how often Potential Colony Habitat is utilized.

Colony Habitat: Potential Colony Habitat that is within the dispersal distance of the species. These areas receive their full original value unless they are further than three home range radii from a core area. These distant areas receive a value of low since there is a low probability that they will be utilized regularly.

Potential Day to Day Habitat: an area of high or medium quality habitat less than one home range, or habitat of low quality of any size. This piece of habitat alone is too small or of inadequate quality to be Core Habitat.

Day to Day Habitat: Potential Day to Day Habitat that is close enough to Core or Colony Habitat can be utilized by individuals moving out from those areas on a day to day basis. The grid

cell must be within Day to Day Distance of Core or Colony Habitat.

Other Habitat: contiguous areas of low value habitat larger than two home ranges in size, including small areas of high and medium quality habitat that may be imbedded in them, are included as usable habitat by the species. Such areas may act as “sinks” because long-term reproduction may not match mortality.

The table below indicates the specific distances and areas assumed by this model.

Distance variables:	Meters	Feet
Dispersal Distance	1,926	6,320
Day to Day Distance/ Home Range radius	80	263

Area variables:	Hectares	M ²	Acres	Ft ²
Home Range	2.02	20,235	5	217,800
Core Habitat	4.04	40,470	10	435,600

Application of the Model

A copy of the ARC/INFO AML can be found in Appendix 1. The steps carried out by the macro are as follows:

1. **Determine Core Habitat:** this is done by first converting all medium quality habitat to high quality habitat and removing all low value habitat. Then contiguous areas of habitat are grouped into regions. The area of each of the regions is determined. Those large enough (two home ranges) are maintained in the Core Habitat coverage. If no Core Habitat is identified then the model will indicate no suitable

habitat in the study area.

2. **Identify Potential Colony Habitat:** using the coverage from Step 1, determine which regions are one to two home ranges in size. These are Potential Colonies.
3. **Identify Potential Day Use Habitat:** using the coverage derived in Step 1, determine which areas qualify as Potential Day to Day Habitat.
4. **Calculate the Cost Grid:** since it is presumed to be more difficult for animals to travel through unsuitable habitat than suitable habitat we use a cost grid to limit travel based on habitat suitability. The cost to travel is one for high or medium quality habitat. This means that to travel 1 m through this habitat costs 1 m of Dispersal Distance. The cost to travel through low quality habitat is two and unsuitable habitat costs four. This means that to travel 1 m through unsuitable habitat costs the species 4 m of Dispersal Distance.
5. **Calculate the Cost Distance Grid:** a cost distance grid containing the minimum cost to travel from each grid cell to the closest Core Habitat is then calculated using the Cost Grid (Step 4) and the Core Habitat (Step 1).
6. **Identify Colony Habitat:** based on the Cost Distance Grid (Step 5), only Potential Colony Habitat within the Dispersal Distance of the species to Core Habitat is retained. Colonies are close enough if **any** cell in the Colony is within the Dispersal Distance from Core Habitat. The suitability of any Colony located further than three home range radii from a Core Habitat is changed to low since it is unlikely it will be utilized regularly.
7. **Create the Core + Colony Grid:** combine the Core Habitat (Step 1) and the Colony Habitat (Step 6) and calculate the cost to travel from any cell to Core or Colony Habitat. This is used to determine which Potential Day to Day Habitat could be utilized.
8. **Identify Day to Day Habitat:** grid cells of Day to Day Habitat are only accessible to the species if they are within Day to Day Distance from the edge of the nearest Core or Colony Habitat. Add these areas to the Core + Colony Grid (Step 7).
9. **Add Other Habitat:** large areas (two home ranges in size) of low value habitat, possibly with small areas of high and medium habitat imbedded in them may be utilized, although marginally. Add these areas back into the Core + Colony + Day to Day Grid (Step 8), if any exist, to create the grid showing areas that will potentially be utilized by the species. Each grid cell contains a one if it is utilized and a zero if it is not.

10. **Restore Values:** all areas that have been retained as having positive habitat value receive their original geometric mean value from the original geometric value grid (see *Cover component* section) with the exception of distant colonies. Distant colonies (colonies more than three home range radii distant) have their value reduced to low because of the low likelihood of utilization.

Problems with the Approach

Cost.

The cost to travel across low suitability and unsuitable habitat is not known. It is likely that it is quite different for different species. This model incorporates a reasonable guess for the cost of movement. A small bird will cross unsuitable habitat much more easily than a small mammal. To some extent differences in vagility between species is accounted for by different estimates of dispersal distances.

Dispersal distance.

The distance animals are willing to disperse from their nest or den site is not well understood. We have used distances from studies of the species or similar species when possible, otherwise first approximations are used. More research is urgently needed on wildlife dispersal.

Day to day distance.

The distance animals are willing to travel on a day to day basis to use distant resources has not been quantified for most species. This issue is less of a concern than dispersal distance since the possible distances are much more limited, especially with small mammals, reptiles, and amphibians. Home range size is assumed to be correlated with this coefficient.

SOURCES OF OTHER MODELS

No other habitat models for mountain chickadee were found.

REFERENCES

- Airola, D.A. 1988. Guide to the California Wildlife Habitat Relationship System. Calif. Dept. of Fish and Game. Sacramento, California. 74 pp.
- Airola, D.A., and R.H. Barrett. 1985. Foraging and habitat relationships of insect-gleaning birds in a Sierra Nevada mixed-conifer forest. *Condor* 87:205-216.
- Beedy, E.C. 1981. Bird communities and forest structure in the Sierra Nevada of California. *Condor* 83:97-105.
- Bevier, L.R. 1988. Mountain Chickadee in J. Farrand, ed. The Audubon Society master guide to birding. Vol. 3, warblers to sparrows. 399 pp.
- Bock, C.E., and J.F. Lynch. 1970. Breeding bird populations of burned and unburned conifer forest in the Sierra Nevada. *Condor* 72:182-189.
- Brennan, L.A. 1989. Comparative use of forest resources by chestnut-backed and mountain chickadees in the western Sierra Nevada. Ph.D. Thesis., Univ. of California, Berkeley, California. 154 pp.
- Dahlsten, D.L., and W.A. Copper. 1979. The use of nesting boxes to study the biology of the mountain chickadee (*Parus gambeli*) and its impact on selected forest insects, pages 217-260 in J. G. Dickson, R.N. Conner, R.R. Fleet, J.C. Kroll, and J.A. Jackson, eds. The role of insectivorous birds in forest ecosystems. Academic Press, New York. 381 pp.
- Dixon, K.L., and J.D. Gilbert. 1964. Altitudinal migration in the mountain chickadee. *Condor* 66:61-64.
- Garrett, K., and J. Dunn. 1981. Birds of southern California. Los Angeles Audubon Soc. 408 pp.
- Grinnell, J., and A.H. Miller. 1944. The distribution of the birds of California. Pac. Coast Avifauna No. 27. 608 pp.
- Grinnell, J., and T.I. Storer. 1924. Animal life in the Yosemite. Univ. of California Press, Berkeley, California. 752 pp.
- Grundel, R., and D.L. Dahlsten. 1990. The feeding ecology of mountain chickadees (*Parus gambeli*): patterns of arthropod prey delivery to nestling birds. *Can. J. Zool.* 69:1793-1804.
- Harrison, C. 1978. A field guide to the nests, eggs and nestlings of North American birds. W. Collins Sons and Co., Cleveland, Oh. 416 pp.
- Hill, G.B., and M.R. Lein. 1988. Ecological relations of sympatric black-capped and mountain

chickadees in southwestern Alberta. *Condor* 90:875-884.

Mayer, K.E., and W.F. Laudenslayer, Jr. eds. 1988. A guide to wildlife habitats of California. Calif. Dept. of Fish and Game, Sacramento, California. 166 pp.

McCaskie, G., P. De Benedictis, R. Erickson, and J. Morlan. 1979. Birds of southern California, an annotated field list. 2nd. ed. Golden Gate Audubon Soc., Berkeley, California. 84 pp.

Morrison, M.L., I.C. Timossi, K.A. With, and P.N. Manley. 1985. Use of tree species by forest birds during winter and summer. *J. Wildl. Manage.* 49(4):1098-1102.

Morrison, M.L., K.A. With, I.C. Timossi, and K.A. Milne. 1987. Composition and temporal variation of flocks in the Sierra Nevada. *Condor* 89:739-745.

Morrison, M.L., D.L. Dahlsten, S.M. Tait, R.C. Heald, K.A. Milne, and D.L. Rowney. 1989. Bird foraging on incense-cedar and incense-cedar scale during winter in California. USDA, For. Ser. Pac. SW For. Range Expt. Stat., Research Paper PSW-195. 10 pp.

Raphael, M.G., and M. White. 1984. Use of snags by cavity-nesting birds in the Sierra Nevada. *Wildlife Monographs* 86:1-66.

Zeiner, D.C., W.F. Laudenslayer, Jr., K.E. Mayer, and M. White, eds. 1990. California's wildlife. Vol. 2 Birds. California Dept. of Fish and Game, Sacramento, California. 732 pp.

APPENDIX 1: Mountain Chickadee Macro

```
/*      MOUNTAIN CHICKADEE

/* mtcmodel.aml - This macro creates an HSI coverage for the
/*      Mountain Chickadee.

/* Version: Arc/Info 6.1 (Unix), GRID-based model.

/* Authors: Irene Timossi, Sarah Miller, Wilde Legard,
/*      and Reginald H. Barrett
/*      Department of Forestry & Resource Management
/*      University of California, Berkeley

/* Revision: 2/10/95

/* -----

/* convert .ID to uppercase for info manipulations

&setvar .ID [translate %.ID%]

/* Start Grid

grid

/*

&type (1) Initializing Constants...

/* Homerange: the size of the species' homerange.

/* DayPay: The amount the species is willing to pay traveling on
/* a day-to-day basis. Used to determine the area utilized on a
/* day-to-day basis.

/* DispersePay: Distance traveled when dispersing. The amount
/* the animal is willing to pay when dispersing from a core area.

/* High: The value in the WHR grid which indicates high quality habitat.

/* Medium: The value in the WHR grid which indicates medium quality habitat.

/* Low: The value in the WHR grid which indicates low quality habitat.

/* None: The value in the WHR grid which indicates habitat of no value.

/* SpecCode: The WHR code for the species

/* AcreCalc: The number needed to convert square units
/*      (feet or meters) to acres.
```

```
&setvar SpecCode = B356
```

```
&if %.Measure% = Meters &then
```

```
&do
```

```
&setvar Homerange    = 20235
```

```
&setvar DayPay        = 80
```

```
&setvar DispersePay   = 1926
```

```
&setvar AcreCalc      = 4047
```

```
&end
```

```
&else
```

```
&if %.Measure% = Feet &then
```

```
&do
```

```
&setvar Homerange    = 217800
```

```
&setvar DayPay        = 263
```

```
&setvar DispersePay   = 6320
```

```
&setvar AcreCalc      = 43560
```

```
&end
```

```
&else
```

```
&do
```

```
&type Measurement type incorrect, check spelling.
```

```
&type Only Meters and Feet are correct.
```

```
&goto &BADEND
```

```
&end
```

```
&setvar High          = 3
```

```
&setvar Medium        = 2
```

```
&setvar Low           = 1
```

```
&setvar None          = 0
```

```
/* The following global variables are declared in the menu:
```

```
/* .WHRgrid (WHR grid name): the name of the grid containing all
```

```
/* the WHR information.
```

```
/* .Bound (Boundary grid name): the grid containing only the
```

```
/* boundary of the coverage. All cells inside the boundary
```

```
/* have a value of 1. All cells outside the boundary must
```

```
/* have a value < 1.
```

```
/* .ID (Identifier): a 1 to 4 character code used to identify
```

```
/* the files produced by this program. You may prefer
```

```
/* to use an abbreviation of the species' common name
```

```
/* (e.g. use `fis1` for fisher).
```

```
/* .SizeOfCell (Cell size): the size (width) of the cells
```

```
/* used in the coverage grids. All grids used in the
```

```
/* analysis must have the same cell size.
```

```
/* .Measure: the units the coverage is measured in (feet or meters).
```


&type (2) Creating working grid of geometric means...

```
/* Create a Geometric Means grid (Geom) for the species by
/* copying these values from the WHR grid.
```

```
Geom = %.WHRgrid%.%SpecCode%_G
```

```
/*
```

&type (3) Changing %Medium% value cells to %High% value for Merge grid...

```
/* Create a grid (Merge) merging Medium and High
/* value cells from the Geometric mean grid (Geom),
/* while leaving the value of other cells (Low and None) unchanged.
/* Merge by changing the value of all medium cells to High.
```

```
Merge = con(Geom == %Medium%,%High%,Geom)
```

```
/*
```

&type (4) Converting Merge grid zones into a Region grid...

```
/* Convert the zones of the merge grid (Merge) into
/* unique regions (Region). These will be used later
/* to create core, colony, and day-to-day areas.
```

```
Region = regiongroup(Merge,#,EIGHT)
```

```
/*
```

&type (5) Calculating the area of Region grid zones...

```
/* Calculate the area of the zones (ZoneArea) on the region
/* grid (Region).
```

```
ZoneArea = zonalarea(Region)
```

```
/*
```

&type (6) Creating a Core Area grid...

```
/* Extract areas from the zonal area grid (ZoneArea)
/* suitable for core areas (Core). Core areas are defined
/* as the Medium+High zones in the merge grid (Merge)
/* with an area of at least two home ranges (%Homerange%).
/* Set their value = 1.
```

```
if (Merge == %High% and ZoneArea >= %Homerange% * 2)
  Core = 1
endif
```

```
&if not [exists Core -vat] &then  
  &goto END
```

```
/*
```

```
&type (7) Creating a Colony grid...
```

```
/* Extract areas from the zonal area grid (zoneArea)  
/* possibly suitable for colonization (ColTemp).  
/* Colony areas are defined as Low or Medium+High zones  
/* in the Merge grid (Merge) with an area of between one  
/* and two home ranges (%Homerange%). Set their value = 1.
```

```
/* Then set all nodata values in the grid to zero (Colony).
```

```
docell
```

```
  if (Merge == %High%)  
    if (ZoneArea > %Homerange% and ZoneArea < %Homerange% * 2)  
      ColTemp = 1  
    endif  
  endif  
end
```

```
Colony = con(isnull(ColTemp),0,ColTemp)
```

```
/*
```

```
&type (8) Creating a Day-to-Day Use grid...
```

```
/* Create a grid based on the values in the zonal  
/* area grid (ZoneArea) and merge grid (Merge)  
/* suitable for day-to-day use (DayToDay). Day-to-day use  
/* areas are defined as Low if the area is less than two  
/* homeranges in size or Medium+High zones in the  
/* merge grid (Merge) with an area of less than one home  
/* range (%Homerange%). Set their value = 1.
```

```
if ((Merge > %Low% and ZoneArea <= %Homerange%) or ~  
    (Merge == %Low% and ZoneArea < %Homerange% * 2))  
  DayToDay = 1  
else  
  DayToDay = 0  
endif
```

```
/*
```

```
&type (9) Creating a Cost Grid based on habitat value...
```

```
/* Using the merge grid (Merge), create a cost grid (Cost)  
/* based on the habitat-value. Cost represents the relative  
/* resistance a species has to moving across different quality
```

```

/*  habitat:  Habitat-value  Cost
/*          None           4
/*          Low            2
/*          Medium+High    1

```

```

if (Merge == %None%)
    Cost = 4
else if (Merge == %Low%)
    Cost = 2
else if (Merge == %High%)
    Cost = 1
endif

```

```

/*

```

&type (10) Calculating cost to travel from Core Areas...

```

/*  Calculate the cost to travel the distance (CostDist)
/*  from the nearest core area source (Core) using the cost
/*  grid (Cost).
/*

```

```

CostDist = CostDistance(Core, Cost)

```

```

/*

```

&type (11) Calculating which Colony areas are Cost Effective...

```

/*  If Colony Areas exist...
/*  Find the areas in the Colony grid (Colony) that could
/*  be colonized from the core areas:

/*  Assign costs to all cells in the Colony areas (Colony)
/*  from the Cost grid (CostDist). Zero surrounding NODATA areas.

/*  Make each colony a separate zone (ZoneReg) using
/*  the regiongroup command.

/*  Use zonalmin to find the minimum cost to arrive at each
/*  colony (ZoneMin).

/*  Set all NODATA cells to zero in ZoneMin to produce
/*  ColZer1.

/*  To find out which of the potential colonies can be utilized,
/*  determine which have a cost that is equal to or less than
/*  DispersePay. If the cost to get to a colony is less than
/*  or equal to DispersePay, keep it in grid Col.

/*  Fill the null value areas in Col with zeros to create ColZer2

```

```

&if not [exists ColTemp -vat] &then
    &goto SkipColony

ColDist = con(Colony > 0, CostDist, 0)
ZoneReg = regiongroup(Colony, #, EIGHT)
ZoneMin = zonalmin(ZoneReg, ColDist)
ColZer1 = con(isnull(ZoneMin), 0, ZoneMin)

if (ColZer1 <= %DispersePay% and ColZer1 > 0)
    Col = Colony
else
    Col = Core
endif

ColZer2 = con(isnull(Col), 0, Col)

/*

&type (12) Creating Core + Colony grid...

/* If colonies exist....
/* Create a grid (ColCore) that combines the core
/* (Core) and colony (Colony) grids.
/* This grid will be used to analyze day-to-day use.

if (Colony == 1)
    ColCore = 1
else
    ColCore = Core
endif

&label SkipColony

&type (13) Calculate cost to travel from Core and Colony Areas...

/* If colonies exist...
/* Calculate the cost to travel the distance (CostDis2)
/* from the nearest core or colony area source (ColCore).
/* Otherwise just copy the CostDist grid to use for Day-to-Day
/* analysis.

&if not [exists ColTemp -vat] &then
    CostDis2 = CostDist
&else CostDis2 = CostDistance(ColCore, Cost)

/*

&type (14) Calculating which Day-to-Day areas are Cost Effective...

```

```

/* This step adds the utilized Day-to-Day cells to the
/* Core + Colony Area grid (ColZer2) to produce the
/* Day1 grid.

/* Use the Core + Colony Cost grid (CostDis2) to find out
/* what can actually be used day-to-day (any cell with
/* a cost of DayPay or less).

/* Retain any cell in the Day-to-Day grid (DayToDay) with
/* a cost less than or equal to DayPay and greater than zero.

/* If the Distance-Cost grid (CostDis2) = 0,
/* it is part of the Core or Colony Area and
/* should get its value from Core + Colony Area
/* grid (ColZer2).

```

```

&if [exists ColTemp -vat] &then
  &do
    if (CostDis2 <= %DayPay% and CostDis2 > 0)
      Day1 = DayToDay
    else
      Day1 = ColZer2
    endif
  &end
&else
  &do
    if (CostDis2 <= %DayPay% and CostDis2 > 0)
      Day1 = DayToDay
    else
      Day1 = Core
    endif
  &end

```

```

/*

```

```

&type (15) Finding Other Areas That May Be Utilized....

```

```

/* This step picks up any large low value areas and any small
/* medium or high value polygons that are imbedded
/* in them.

/* First find any areas that are not currently in the included
/* set (Day1Z) but are in the original geometric mean coverage (geom)
/* set Other to 1 where Day1Z = 0.

/* if Other is all nodata, create the All coverage from
/* the Day1Z coverage.

/* Split all Other areas into separate regions (OthReg)

/* Calculate the area of the regions (OthArea).

```

```
/* Keep any region in OthArea with an area > 2 homeranges (Util).
```

```
/* Change any null values in Util to zeros (OthZero).
```

```
/* Add these areas to the Day1 coverage to create All
```

```
Day1Z = con(isnull(Day1),0,Day1)
```

```
if ((Day1Z < 1) and (Geom > 0))
```

```
    Other = 1
```

```
endif
```

```
&if not [exists Other -vat] &then
```

```
    All = Day1Z
```

```
&else
```

```
    &do
```

```
        OthReg = regiongroup(Other,#,EIGHT)
```

```
        OthArea = zonalarea(OthReg)
```

```
        if (OthArea >= %Homerange% * 2)
```

```
            Util = 1
```

```
        else
```

```
            Util = 0
```

```
        endif
```

```
OthZero = con(isnull(Util),0,Util)
```

```
if (OthZero == 1)
```

```
    All = OthZero
```

```
else
```

```
    All = Day1Z
```

```
endif
```

```
&end
```

```
/*
```

```
&type (16) Creating a Value grid...
```

```
/* For any cell in All that has a value of 1, store the suitability
```

```
/* value from the Geometric mean grid (Geom) to the Value grid.
```

```
/* Other cells inside the boundary (%.Bound%) get a value of 0.
```

```
/*
```

```
if (All == 1)
```

```
    Value = Geom
```

```
else if (%.Bound% == 1)
```

```
    Value = 0
```

```
endif
```

```
/*
```

```
&type (17) Creating an HSI grid...
```

```
/* if Colonies exist....
```

```
/* For any cell that was part of a colony that is further than
```

```
/* 3 times the HR radius (DayPay) away from a core area, set the suitability
```

```
/* to Low. Distant colonies lose value because of their small size.
```

```
/* This step produces grid Collow.
```

```
/* Set all NODATA values in Collow to zero in ColZer3.
```

```
/* Find any day-to-day use areas (DayToDay) that are being
```

```
/* utilized (ColZer3). If they are further than four homeranges
```

```
/* from a core area (CostDist), they are utilized from a distant
```

```
/* colony and their value will be decreased to Low in Day2.
```

```
/* Then change nulls to zero in ValZero
```

```
/* Keep all data within the boundary; call this final grid HSI.
```

```
&if [exists ColTemp -vat] &then
```

```
&do
```

```
if (ColZer1 >= %DayPay% * 3)
```

```
Collow = %Low%
```

```
else
```

```
Collow = Value
```

```
endif
```

```
ColZer3 = con(isnull(Collow),0,Collow)
```

```
if ((CostDist > %DayPay% * 4) and (ColZer3 > 0) and ~  
(DayToDay == 1))
```

```
Day2 = 1
```

```
else
```

```
Day2 = ColZer3
```

```
endif
```

```
&end
```

```
&else
```

```
Day2 = Value
```

```
valzero = con(isnull(Day2),0,Day2)
```

```
if (%.Bound% == 1)
```

```
%ID%hsi = valzero
```

```
endif
```

```
/*
```

```
&type (18) Quitting from GRID and adding the acres field.....
```

```
/* Quit from GRID (Q), then run additem to add an acre item to
/* the HSI grid vat file (%ID%HSI.vat). Reindex on value when done.
```

Q

```
additem %.ID%hsi.vat %.ID%hsi.vat acres 10 10 i
indexitem %.ID%hsi.vat value
```

```
/*
```

&type (19) Calculating acres.....

```
/* Use INFO to calculate the acreage field: Multiply the number
/* of cells by the cell size squared and divide by the number of
/* square meters per acre (4047). Reindex on value when done.
```

```
&data arc info
arc
select %.ID%hsi.VAT
CALC ACRES = ( COUNT * %.SizeOfCell% * %.SizeOfCell% ) / %AcreCalc%
Q STOP
```

&END

```
indexitem %.ID%hsi.vat value
```

```
/*
```

&type (20) Killing all intermediate coverages before ending macro...

```
/* &goto OKEND
```

grid

```
kill Geom
kill Merge
kill Region
kill ZoneArea
kill Core
kill ColTemp
kill Colony
kill DayToDay
kill Cost
kill CostDist
kill ColDist
kill ZoneReg
kill ZoneMin
kill ColZer1
kill Col
kill ColZer2
kill ColCore
kill CostDis2
```


kill Day1
kill Day1Z
kill Other
kill OthReg
kill OthArea
kill Util
kill OthZero
kill All
kill Value
kill Collow
kill ColZer3
kill Day2
kill valzero

q

&goto OKEND

&label END
&type **
&type **
&type NO CORE AREAS EXIST, EXITING MACRO
&type **
&type **

kill Core
kill Region
kill ZoneArea
kill Merge
kill Geom

quit

&label OKEND
&label BADEND

&type ----- All done! -----

&return